

6. Modularizarea aplicațiilor prin utilizarea subprogramelor

Noțiunea de subprogram (procedură sau funcție) a fost concepută cu scopul de a grupa o mulțime de comenzi *SQL* cu instrucțiuni procedurale, pentru a construi o unitate logică de tratare a unei anumite probleme. În general, procedurile sunt folosite pentru a realiza o acțiune, iar funcțiile pentru a calcula o valoare.

Unitățile de program care pot fi create în *PL/SQL* sunt:

- subprograme locale (definite în partea declarativă a unui bloc *PL/SQL* sau a unui alt subprogram);
- subprograme independente (stocate în baza de date și considerate obiecte ale acesteia);
- subprograme împachetate (definite într-un pachet care încapsulează proceduri și funcții).

Procedurile și funcțiile stocate sunt unități de program *PL/SQL* apelabile (compilate), care există ca obiecte în schema bazei de date *Oracle*. Recuperarea unui subprogram (în cazul unei corecții) nu cere recuperarea întregii aplicații. Subprogramul, încărcat în memorie pentru a fi executat, poate fi partajat între aplicațiile care îl solicită.

Este important de făcut distincție între procedurile stocate și procedurile locale (declarate și folosite în blocuri anonime).

- Procedurile și funcțiile stocate, care sunt compilate și stocate în baza de date, nu mai trebuie să fie compilate la fiecare execuție, în timp ce procedurile locale sunt compilate de fiecare dată când este executat blocul care le conține.
- Procedurile declarate și apelate în blocuri anonime sunt temporare (ele nu mai există după ce blocul anonim a fost executat complet). O procedură stocată (creată cu *CREATE PROCEDURE* sau conținută într-un pachet) este permanentă, în sensul că ea poate fi invocată de un fișier *SQL*Plus*, un subprogram *PL/SQL* sau un declanșator.
- Procedurile și funcțiile stocate pot fi apelate din orice bloc, de către utilizatorul care are privilegiul *EXECUTE* asupra acestora, în timp ce procedurile și funcțiile locale pot fi apelate numai din blocul care le conține.

Când este creat un subprogram stocat, utilizând comanda *CREATE*, subprogramul este depus în dicționarul datelor. Este depus atât textul sursă, cât și forma compilată (*p-code*). Atunci când subprogramul este apelat, *p-code* este citit de pe disc, este depus în *shared pool*, unde poate fi accesat de mai mulți

utilizatori și este executat dacă este necesar. El va părăsi *shared pool* conform algoritmului *LRU* (*least recently used*).

Pachetul *DBMS_SHARED_POOL* permite păstrarea de obiecte în *shared pool*. Dacă un obiect este gestionat în această manieră, el va părăsi *shared pool* doar dacă aceasta se cere explicit. Pentru a păstra în *shared pool* subprograme, pachete, declanșatori, cursoare, clase *Java*, tipuri obiect, comenzi *SQL*, secvențe este utilizată procedura *DBMS_SHARED_POOL.KEEP*. Unica modalitate de a șterge un obiect păstrat în *shared pool*, fără a reporni baza, este cu ajutorul procedurii *DBMS_SHARED_POOL.UNKEEP*.

Subprogramele se pot declara în blocuri *PL/SQL*, în alte subprograme sau în pachete, dar la sfârșitul secțiunii declarative. La fel ca blocurile *PL/SQL* anonime, subprogramele conțin o parte declarativă, o parte executabilă și, opțional, o parte de tratare a erorilor. Partea declarativă conține declarații de tipuri, cursoare, constante, variabile, excepții și subprograme imbricate. Partea executabilă conține instrucțiuni care asignează valori, controlează execuția programului și prelucrează datele. Cea de-a treia parte se ocupă cu tratarea excepțiilor apărute în timpul execuției subprogramului.

Crearea subprogramelor stocate

Principalele etape pentru crearea unui subprogram stocat sunt următoarele:

- se editează subprogramul (*CREATE PROCEDURE* sau *CREATE FUNCTION*) și se salvează într-un *script file SQL*;
- se încarcă și se execută acest *script file*, se compilează codul sursă, se obține *p-code* (subprogramul este creat);
- se utilizează comanda *SHOW ERRORS* pentru vizualizarea eventualelor erori la compilare (comanda *CREATE PROCEDURE* sau *CREATE FUNCTION* depune codul sursă în dicționarul datelor chiar dacă subprogramul conține erori la compilare);
- se execută subprogramul pentru a realiza acțiunea dorită (de exemplu, procedura poate fi executată fie utilizând comanda *EXECUTE* din *iSQL*Plus*, fie invocând-o dintr-un bloc *PL/SQL*).

Când este apelat subprogramul, motorul *PL/SQL* execută *p-code*.

Dacă există erori la compilare și se fac corecțiile corespunzătoare, atunci este necesară fie comanda *DROP PROCEDURE* (respectiv *DROP FUNCTION*), fie sintaxa *OR REPLACE* în cadrul comenzii *CREATE*.

Atunci când este apelată o procedură *PL/SQL*, server-ul *Oracle* parcurge anumite etape care vor fi detaliate în cele ce urmează.

- Verifică dacă utilizatorul are privilegiul să execute procedura (fie pentru că el a creat procedura, fie pentru că i s-a acordat acest privilegiu).
- Verifică dacă procedura este prezentă în *shared pool*. Dacă este prezentă va fi executată, altfel va fi încărcată de pe disc în *database buffer cache*.
- Verifică dacă starea procedurii este validă (*VALID*) sau invalidă (*INVALID*). Starea unei proceduri *PL/SQL* este invalidă, fie pentru că au fost detectate erori la compilarea procedurii, fie pentru că structura unui obiect s-a schimbat de când procedura a fost executată ultima oară. Dacă starea este invalidă atunci procedura este recompilată automat. Dacă nici o eroare nu a fost detectată, atunci va fi executată noua versiune a procedurii.
- Dacă procedura aparține unui pachet atunci toate procedurile și funcțiile pachetului sunt de asemenea încărcate în *database buffer cache* (dacă nu erau deja acolo). Dacă pachetul este activat pentru prima oară într-o sesiune, atunci *server*-ul va executa blocul de inițializare al pachetului.

Proceduri *PL/SQL*

O procedură *PL/SQL* este un program independent care se află compilat în schema bazei de date *Oracle*. Când procedura este compilată, identificatorul acesteia (stabilit prin comanda *CREATE PROCEDURE*) devine un nume de obiect în dicționarul datelor. Tipul obiectului este *PROCEDURE*.

Sintaxa generală pentru crearea unei proceduri este următoarea:

```
[CREATE [OR REPLACE] ] PROCEDURE nume_procedură
                                [ (parametru [, parametru ...] ) ]
    [AUTHID {DEFINER | CURRENT_USER}]
    {IS | AS}
    [PRAGMA AUTONOMOUS_TRANSACTION];
    [declarații locale]
BEGIN
    partea_executabilă
[EXCEPTION
    partea_de_tratare_a_excepțiilor]
END [nume_procedură];
```

Parametrii au următoarea formă sintactică:

```
nume_parametru [ {IN | OUT [NOCOPY] | IN OUT [NOCOPY] } ]
                tip_de_date [ {:= | DEFAULT} expresie]
```

Comanda *CREATE* permite ca procedura să fie stocată în baza de date. Când procedurile sunt create folosind clauza *CREATE OR REPLACE*, ele vor fi stocate în baza de date în formă compilată, formă care permite execuția mai rapidă a acestora. Dacă procedura există, atunci clauza *OR REPLACE* va avea ca efect ștergerea procedurii și înlocuirea acesteia cu noua versiune. Dacă procedura există, iar opțiunea *OR REPLACE* nu este prezentă, atunci comanda *CREATE* va returna eroarea „ORA-00955: Name is already used by an existing object”.

Clauza *AUTHID* specifică faptul că procedura stocată se execută cu drepturile proprietarului (implicit) sau ale utilizatorului curent. De asemenea, această clauză precizează dacă referințele la obiecte sunt rezolvate în schema proprietarului procedurii sau în cea a utilizatorului curent.

Clauza *PRAGMA AUTONOMOUS TRANSACTION* anunță compilatorul *PL/SQL* că această procedură este autonomă (independentă). Tranzacțiile autonome permit suspendarea tranzacției principale, executarea unor instrucțiuni *SQL*, permanentizarea sau anularea acestor operații și continuarea tranzacției principale.

Parametrii formali (variabile declarate în lista parametrilor specificației subprogramului) pot să fie de tipul *%TYPE*, *%ROWTYPE* sau de un tip explicit, fără specificarea dimensiunii.

Exemplu:

Să se creeze o procedură stocată care micșorează cu o valoare dată (*cant*) polițele de asigurare emise de firma *ASIROM*.

```
CREATE OR REPLACE PROCEDURE mic (cant IN NUMBER) AS
BEGIN
    UPDATE politaasig
    SET     valoare = valoare - cant
    WHERE  firma = 'ASIROM';
END;
/
```

Dacă în subprograme se execută operații de reactualizare și există declanșatori relativ la aceste operații care nu trebuie să se execute, atunci înainte de apelarea subprogramului declanșatorii trebuie dezactivați, urmând ca ei să fie reactivați după ce s-a terminat execuția subprogramului.

De exemplu, în problema prezentată anterior ar trebui dezactivați declanșatorii referitori la tabelul *politaasig*, apelată procedura *mic* și în final, reactivați acești declanșatori.

```
ALTER TABLE politaasig DISABLE ALL TRIGGERS;
EXECUTE mic(10000)
```

5

```
ALTER TABLE politaasig ENABLE ALL TRIGGERS;
```

Exemplu:

Să se creeze o procedură locală prin care se inserează informații în tabelul *editata_de*.

```
DECLARE
    PROCEDURE editare
        (v_cod_sursa editata_de.cod_sursa%TYPE,
         v_cod_autor  editata_de.cod_autor%TYPE)
IS
    BEGIN
        INSERT INTO editata_de
        VALUES (v_cod_sursa,v_cod_autor);
    END;
BEGIN
    ...
    editare(75643, 13579);
    ...
END;
/
```

Procedurile stocate pot fi apelate:

- din corpul altei proceduri sau al unui declanșator;
- interactiv, de către utilizator, folosind un instrument *Oracle* (de exemplu, *iSQL*Plus*);
- explicit dintr-o aplicație (de exemplu, *Oracle Forms* sau prin utilizarea de precompilatoare).

Apelarea unei proceduri se poate face în funcție de mediul care o solicită:

1) în *SQL*Plus*, prin comanda

EXECUTE *nume_procedură* [(*lista_parametri_actuali*)];

2) în *PL/SQL*, prin invocarea numelui procedurii urmat de lista parametrilor actuali.

Parametrii actuali sunt variabile sau expresii referite în lista parametrilor subprogramului apelant. Ei trebuie să fie compatibili ca tip și număr cu parametrii formali.

Funcții *PL/SQL*

O funcție *PL/SQL* este similară unei proceduri cu excepția că ea trebuie să întoarcă un rezultat. O funcție fără comanda *RETURN* va genera o eroare la compilare.

Când funcția este compilată, identificatorul acesteia devine obiect în dicționarul datelor având tipul *FUNCTION*. Algoritmul din interiorul corpului subprogramului funcție trebuie să asigure faptul că toate traiectoriile sale conduc la comanda *RETURN*. Dacă o traiectorie a algoritmului trimite în partea de tratare a erorilor, atunci *handler*-ul acesteia trebuie să includă o comandă *RETURN*. Orice funcție trebuie să conțină clauza *RETURN* în antet și cel puțin o comandă *RETURN* în partea executabilă.

Sintaxa simplificată pentru scrierea unei funcții este următoarea:

```
[CREATE [OR REPLACE] ] FUNCTION nume_funcție
                                     [ (parametru [, parametru ...] ) ]
    RETURN tip_de_date
    [AUTHID {DEFINER | CURRENT_USER} ]
    [DETERMINISTIC]
    {IS | AS}
    [PRAGMA AUTONOMOUS_TRANSACTION;]
    [declarații locale]
BEGIN
    partea_executabilă
[EXCEPTION
    partea_de_tratare_a_excepțiilor]
END [nume_funcție];
```

Opțiunea *tip_de_date* specifică tipul valorii returnate de funcție, tip care nu poate conține specificații de dimensiune. Dacă totuși sunt necesare aceste specificații, se pot defini subtipuri, iar parametrii și valoarea returnată vor fi declarați de acel subtip.

În interiorul funcției trebuie să apară instrucțiunea *RETURN expresie*, unde *expresie* este valoarea rezultatului furnizat de funcție. Pot să fie mai multe comenzi *RETURN* într-o funcție, dar numai una din ele va fi executată. Comanda *RETURN* (fără o expresie asociată) poate să apară și într-o procedură. În acest caz, ea va avea ca efect saltul la comanda ce urmează instrucțiunii apelante.

Opțiunea *DETERMINISTIC* ajută optimizorul *Oracle* în cazul unor apeluri repetate ale aceleiași funcții, cu aceleași argumente. Ea indică posibilitatea folosirii unui rezultat obținut anterior.

Observații:

- În blocul *PL/SQL* al unei funcții stocate (cel care definește acțiunea

efectuată de funcție) nu pot fi referite variabile *host* sau variabile *bind*.

- O funcție poate accepta unul sau mai mulți parametri, dar trebuie să returneze o singură valoare. Ca și în cazul procedurilor, lista parametrilor este opțională. Dacă subprogramul nu are parametri, parantezele nu sunt necesare la declarare și la apelare.
- O procedură care conține un parametru de tip *OUT* poate fi rescrisă sub forma unei funcții.

Exemplu:

Să se creeze o funcție stocată care determină numărul fotografiilor realizate pe hartie lucioasă, ce au fost achiziționate la o anumită dată.

```
CREATE OR REPLACE FUNCTION  numar_fotografii
        (v_a  IN  fotografie.data_achizitie%TYPE)
RETURN NUMBER AS
    alfa  NUMBER;
BEGIN
    SELECT  COUNT(ROWID)
    INTO    alfa
    FROM    fotografie
    WHERE   hartie = 'lucioasa'
    AND     data_achizitie = v_a;
    RETURN alfa;
END numar_fotografii;
/
```

Dacă apare o eroare de compilare, utilizatorul o va corecta în fișierul editat și apoi va trimite compilatorului (cu opțiunea *OR REPLACE*) fișierul modificat.

Sintaxa pentru apelul unei funcții este:

```
[ [schema.]nume_pachet.]nume_funcție [@dblink] [(lista_parametri_actuali) ];
```

O funcție stocată poate fi apelată în mai multe moduri. În continuare sunt prezentate trei exemple de apelare.

- 1) Apelarea funcției și atribuirea valorii acesteia într-o variabilă de legătură *SQL*Plus*:

```
VARIABLE val NUMBER
EXECUTE :val := numar_fotografii(SYSDATE)
PRINT val
```

Când este utilizată declarația *VARIABLE* pentru variabilele *host* de tip *NUMBER* nu trebuie specificată dimensiunea, iar pentru cele de tip *CHAR* sau

VARCHAR2 valoarea implicită este 1 sau poate fi specificată o altă valoare între paranteze. *PRINT* și *VARIABLE* sunt comenzi *SQL*Plus*.

2) Apelarea funcției într-o instrucțiune *SQL*:

```
SELECT numar_fotografii(SYSDATE)
FROM dual;
```

3) Apariția numelui funcției într-o comandă din interiorul unui bloc *PL/SQL* (de exemplu, într-o instrucțiune de atribuire):

```
SET SERVEROUTPUT ON
ACCEPT data PROMPT 'dati data achizitionare'
DECLARE
    num        NUMBER;
    v_data     fotografie.data_achizitie%TYPE := '&data';
BEGIN
    num := numar_fotografii(v_data);
    DBMS_OUTPUT.PUT_LINE('numarul fotografiilor
        achizitionate la data ' || TO_CHAR(v_data) || '
este ' || TO_CHAR(num));
END;
/
SET SERVEROUTPUT OFF
```

Exemplu:

Să se creeze o procedură stocată care pentru un anumit gen de fotografie (dat ca parametru) calculează numărul fotografiilor expuse din genul respectiv, numărul de artiști care au creat aceste fotografii, numărul de expoziții în care au fost expuse, precum și valoarea nominală totală a acestora.

```
CREATE OR REPLACE PROCEDURE date_gen_fotografie
    (v_gen fotografie.tip%TYPE) AS
    FUNCTION nr_fotografii (v_gen fotografie.tip%TYPE)
    RETURN NUMBER IS
        v_numar NUMBER(3);
    BEGIN
        SELECT COUNT(*)
        INTO    v_numar
        FROM    fotografie
        WHERE   gen = v_gen;
        RETURN v_numar;
    END nr_fotografii;
    FUNCTION valoare_totala
```



```

        (v_gen fotografie.gen%TYPE)
RETURN NUMBER IS
    v_numar  fotografie.valoare%TYPE;
BEGIN
    SELECT SUM(valoare)
    INTO    v_numar
    FROM    fotografie
    WHERE   gen = v_gen;
    RETURN v_numar;
END valoare_totala;
FUNCTION nr_artisti (v_gen fotografie.gen%TYPE)
RETURN NUMBER IS
    v_numar  NUMBER(3);
BEGIN
    SELECT COUNT(DISTINCT cod_artist)
    INTO    v_numar
    FROM    fotografie
    WHERE   gen = v_gen;
    RETURN v_numar;
END nr_artisti;
FUNCTION nr_expozitii (v_gen fotografie.gen%TYPE)
RETURN NUMBER IS
    v_numar  NUMBER(3);
BEGIN
    SELECT COUNT(DISTINCT s.cod_expozitie)
    INTO    v_numar
    FROM    fotografie f, sala s
    WHERE   f.cod_sala = s.cod_sala
    AND     gen = v_gen;
    RETURN v_numar;
END nr_expozitii;
BEGIN
    DBMS_OUTPUT.PUT_LINE('Numarul    fotografiilor    este
'|| nr_fotografii(v_gen));
    DBMS_OUTPUT.PUT_LINE('Valoarea    fotografiilor    este
'|| valoare_totala(v_gen));
    DBMS_OUTPUT.PUT_LINE('Numarul de artisti este
'|| nr_artisti(v_gen));
    DBMS_OUTPUT.PUT_LINE('Numarul de expozitii este
'|| nr_expozitii(v_gen));

```

```
END date_gen_fotografie;
```

Instrucțiunea *CALL*

O instrucțiune specifică pentru *Oracle9i* este *CALL*, care permite apelarea subprogramelor *PL/SQL* stocate (independente sau incluse în pachete) și a metodelor *Java*.

CALL este o comandă *SQL* care nu poate să apară de sine stătătoare într-un bloc *PL/SQL*. Ea poate fi utilizată în *PL/SQL* doar dinamic, prin intermediul comenzii *EXECUTE IMMEDIATE*. Pentru executarea acestei comenzi, utilizatorul trebuie să aibă privilegiul *EXECUTE* asupra subprogramului. Instrucțiunea poate fi executată interactiv din *SQL*Plus*.

Comanda *CALL* are sintaxa următoare:

```
CALL [schema.] [ {nume_tip_obiect | nume_pachet}. ] nume_subprogram  
[ (lista_parametri_actuali) ] [ @dblink_nume ] [ INTO :variabila_host ]
```

Identificatorul *nume_subprogram* este numele unui subprogram sau al unei metode apelate. Clauza *INTO* este folosită numai pentru variabilele de ieșire ale unei funcții. Dacă lipsește clauza *@dblink_nume*, atunci apelul se referă la baza de date locală, iar într-un sistem distribuit clauza specifică numele bazei de date ce conține subprogramul.

Exemplu:

Sunt prezentate două exemple prin care o funcție *PL/SQL* este apelată din *SQL*Plus*, respectiv o procedură externă *C* este apelată, folosind *SQL* dinamic, dintr-un bloc *PL/SQL*.

```
CREATE OR REPLACE FUNCTION apelfunctie(a IN VARCHAR2)  
  RETURN VARCHAR2 AS  
BEGIN  
  DBMS_OUTPUT.PUT_LINE ('Apel functie cu ' || a);  
  RETURN a;  
END apelfunctie;  
/
```

```
SQL> --apel valid  
SQL> VARIABLE v_iesire VARCHAR2(20)  
SQL> CALL apelfunctie('Salut!') INTO :v_iesire  
Apel functie cu Salut!  
Call completed  
  
SQL> PRINT v_iesire  
v_iesire  
Salut!
```

11

```
DECLARE
  a  NUMBER(7);
  x  VARCHAR2(10);
BEGIN
  EXECUTE IMMEDIATE 'CALL alfa_extern_procedura (:aa,
:xx)' USING a, x;
END;
/
```

Modificarea și suprimarea subprogramelor *PL/SQL*

Pentru a lua în considerare modificarea unei proceduri sau funcții, recompilarea acesteia se face prin comanda:

ALTER {FUNCTION | PROCEDURE} [schema.]nume COMPILE;

Ca și în cazul tabelelor, funcțiile și procedurile pot fi suprimate cu ajutorul comenzii *DROP*. Aceasta presupune eliminarea subprogramelor din dicționarul datelor. *DROP* este o comandă ce aparține limbajului de definire a datelor, astfel că se execută un *COMMIT* implicit atât înainte, cât și după comandă.

Atunci când este șters un subprogram prin comanda *DROP*, automat sunt revocate toate privilegiile acordate referitor la acest subprogram. Dacă este utilizată sintaxa *CREATE OR REPLACE*, privilegiile acordate acestui subprogram rămân aceleași.

Comanda *DROP* are următoarea sintaxă:

DROP {FUNCTION | PROCEDURE} [schema.]nume;

Transferarea valorilor prin parametri

Lista parametrilor unui subprogram este compusă din parametri de intrare (*IN*), de ieșire (*OUT*) sau de intrare/ieșire (*IN OUT*), separați prin virgulă.

Dacă nu este specificat tipul parametrului, atunci implicit acesta este considerat de intrare (*IN*). Un parametru formal cu opțiunea *IN* poate primi valori implicite chiar în cadrul comenzii de declarare. Acest parametru este *read-only* și deci, nu poate fi schimbat în corpul subprogramului. El acționează ca o constantă. Parametrul actual corespunzător poate fi literal, expresie, constantă sau variabilă inițializată.

Un parametru formal cu opțiunea *OUT* este neinițializat și prin urmare, are automat valoarea *null*. În interiorul subprogramului, parametrilor cu opțiunea *OUT* sau *IN OUT* trebuie să li se asigneze o valoare explicită. Dacă nu se atribuie nici o valoare, atunci parametrul actual corespunzător va avea valoarea

null. Parametrul actual trebuie să fie o variabilă, nu poate fi o constantă sau o expresie.

Dacă în timpul execuției procedurii apare o excepție, atunci valorile parametrilor formali cu opțiunile *IN OUT* sau *OUT* nu sunt copiate în valorile parametrilor actuali.

Implicit, transmiterea parametrilor se face prin referință în cazul parametrilor *IN* și prin valoare în cazul parametrilor *OUT* sau *IN OUT*. Dacă pentru realizarea unor performanțe se dorește transmiterea prin referință și a parametrilor *IN OUT* sau *OUT*, atunci se poate utiliza opțiunea *NOCOPY*. Dacă opțiunea *NOCOPY* este asociată unui parametru *IN*, atunci se va genera o eroare la compilare, deoarece acești parametri se transmit de fiecare dată prin referință.

Opțiunea *NOCOPY* va fi ignorată, iar parametrul va fi transmis prin valoare dacă:

- parametrul actual este o componentă a unui tablou indexat (restricția nu se aplică dacă parametrul este întreg tabelul);
- parametrul actual este constrâns prin specificarea unei precizii, a unei mărimi sau prin opțiunea *NOT NULL*;
- parametrul formal și cel actual asociat sunt înregistrări care fie au fost declarate implicit ca variabile contor într-un ciclu *LOOP*, fie au fost declarate explicit prin *%ROWTYPE*, dar constrângerile pe câmpurile corespunzătoare diferă;
- transmiterea parametrului actual cere o conversie implicită a tipului;
- subprogramul este o parte a unui apel de tip *RPC* (*remote procedure call*), iar în acest caz, parametrii fiind transmiși în rețea, transferul nu se poate face prin referință.

Atunci când este apelată o procedură *PL/SQL*, sistemul *Oracle* furnizează două metode pentru definirea parametrilor actuali: specificarea explicită prin nume și specificarea prin poziție.

Exemplu:

Sunt prezentate diferite moduri pentru apelarea procedurii *p1*.

```
CREATE PROCEDURE p1 (a IN NUMBER, b IN VARCHAR2,
                    c IN DATE, d OUT NUMBER) AS ...;

DECLARE
    var_a    NUMBER;
    var_b    VARCHAR2;
    var_c    DATE;
    var_d    NUMBER;
BEGIN
```

13

```
--specificare prin poziție  
p1(var_a,var_b,var_c,var_d);  
--specificare prin nume  
p1(b=>var_b,c=>var_c,d=>var_d,a=>var_a);  
--specificare prin nume și poziție  
p1(var_a,var_b,d=>var_d,c=>var_c);  
END;
```

Dacă este utilizată specificația prin poziție, parametrii care au primit o valoare implicită trebuie să fie plasați la sfârșitul listei parametrilor actuali.

Exemplu:

Fie *proces_data* o procedură care procesează în mod normal data zilei curente, dar care opțional poate procesa și alte date. Dacă nu se specifică parametrul actual corespunzător parametrului formal *plan_data*, atunci acesta va lua automat valoarea dată implicit (data curentă a sistemului).

```
PROCEDURE proces_data(data_in      IN NUMBER,  
                      plan_data    IN DATE := SYSDATE)  
IS ...
```

Următoarele comenzi reprezintă apeluri corecte ale procedurii *proces_data*:

```
proces_data(10);  
proces_data(10,SYSDATE+1);  
proces_data(plan_data=>SYSDATE+1,data_in=>10);
```

O declarație de subprogram (procedură sau funcție) fără parametri este specificată fără paranteze. De exemplu, dacă procedura *react_calc_dur* și funcția *obt_date* nu au parametri, atunci:

```
react_calc_dur;           -- apel corect  
react_calc_dur();        -- apel incorect  
data_mea := obt_date;    -- apel corect
```

Module overload

Două sau mai multe module pot să aibă aceleași nume, dar să difere prin lista parametrilor. Aceste module sunt numite module *overload* (supraîncărcate). Funcția *TO_CHAR* este un exemplu de modul *overload*. Există o singură funcție, *TO_CHAR*, pentru a converti date numerice și calendaristice în date de tip caracter.

În cazul unui apel, compilatorul compară parametrii actuali cu listele parametrilor formali pentru modulele *overload* și execută modulul corespunzător. Toate programele *overload* trebuie să fie definite în același bloc

PL/SQL (bloc anonim, modul sau pachet).

Modulele *overload* pot să apară în programele *PL/SQL* fie în secțiunea declarativă a unui bloc, fie în interiorul unui pachet. Supraîncărcarea subprogramelor nu se poate face pentru funcții sau proceduri stocate, dar este permisă pentru subprograme locale, subprograme care apar în pachete sau pentru metode.

Observații:

- Două subprograme *overload* trebuie să difere, cel puțin prin tipul unuia dintre parametri. Două subprograme nu pot fi *overload* dacă parametrii lor formali diferă numai prin tipurile lor și dacă acestea sunt niște subtipuri care se bazează pe același tip de date.
- Nu este suficient ca lista parametrilor subprogramelor *overload* să difere numai prin numele parametrilor formali.
- Nu este suficient ca lista parametrilor subprogramelor *overload* să difere numai prin tipul acestora (*IN*, *OUT*, *IN OUT*). *PL/SQL* nu poate face diferența (la apelare) între tipurile *IN* și *OUT*.
- Nu este suficient ca funcțiile *overload* să difere doar prin tipul de date returnat (tipul de date specificat în clauza *RETURN* a funcției).
-

Exemplu:

Următoarele subprograme nu pot fi *overload*.

- 1) `FUNCTION alfa(par IN POSITIVE) ...;`
`FUNCTION alfa(par IN BINARY_INTEGER) ...;`
- 2) `FUNCTION alfa(par IN NUMBER) ...;`
`FUNCTION alfa(parar IN NUMBER) ...;`
- 3) `PROCEDURE beta(par IN VARCHAR2) IS...;`
`PROCEDURE beta(par OUT VARCHAR2) IS...;`

Exemplu:

Să se creeze două funcții (locale) cu același nume care să calculeze media valorilor fotografiilor dintr-un anumit gen. Prima funcție va avea un argument reprezentând genul fotografiilor, iar cea de-a doua va avea două argumente, unul reprezentând genul fotografiilor, iar celălalt reprezentând artistul pentru care se calculează valoarea medie (funcția va calcula media valorilor fotografiilor dintr-un anumit gen și care sunt create de un artist).

DECLARE

 medie1 NUMBER(10,2);

 medie2 NUMBER(10,2);

 FUNCTION valoare_medie (v_gen fotografie.tip%TYPE)

15

```
        RETURN NUMBER IS
    medie NUMBER(10,2);
BEGIN
    SELECT AVG(valoare)
    INTO    medie
    FROM    fotografie
    WHERE   gen = v_gen;
    RETURN medie;
END;
FUNCTION valoare_medie
    (v_gen    fotografie.gen%TYPE,
     v_artist  fotografie.cod_artist%TYPE)
    RETURN NUMBER IS
    medie NUMBER(10,2);
BEGIN
    SELECT AVG(valoare)
    INTO    medie
    FROM    fotografie
    WHERE   gen = v_gen AND cod_artist = v_artist;
    RETURN medie;
END;
BEGIN
    mediel := valoare_medie('natura');
    DBMS_OUTPUT.PUT_LINE('Media valorilor fotografiilor
din natura este ' || mediel);
    medie2 := valoare_medie('natura ', 1);
    DBMS_OUTPUT.PUT_LINE('Media valorilor fotografiilor
din natura create de artistul 1 este ' || medie2);
END;
```

Procedură versus funcție

După cum am mai subliniat, în general, procedura este utilizată pentru realizarea unor acțiuni, iar funcția este folosită pentru calculul unei valori.

Pot fi marcate câteva deosebiri esențiale între funcții și proceduri.

- Procedura se execută ca o comandă *PL/SQL*, iar funcția se invocă în cadrul unei expresii.
- Procedura poate returna (sau nu) una sau mai multe valori, iar funcția trebuie să returneze o singură valoare.
- Procedura nu trebuie să conțină clauza *RETURN expresie*, iar funcția

trebuie să conțină această opțiune.

De asemenea, pot fi remarcate câteva elemente esențiale, comune atât funcțiilor cât și procedurilor. Ambele pot:

- accepta valori implicite;
- conține secțiuni declarative, executabile și de tratare a erorilor;
- utiliza specificarea prin nume sau poziție a parametrilor;
- accepta parametri *NOCOPY*.

Tratarea excepțiilor

Dacă apare o eroare într-un subprogram, atunci este declanșată o excepție. Dacă excepția este tratată în subprogram, atunci blocul se termină și controlul trece la secțiunea de tratare a erorilor din subprogramul respectiv. Dacă nu există o tratare a excepției în subprogram, atunci controlul trece la programul apelant de nivel imediat superior (în partea de tratare a erorilor), respectând regulile de propagare a excepțiilor. În acest caz, valorile parametrilor de tip *OUT* sau *IN OUT* nu sunt returnate parametrilor actuali. Aceștia vor avea aceleași valori, ca și cum subprogramul nu ar fi fost apelat.

Recursivitate

Recursivitatea este o tehnică importantă pentru simplificarea modelării algoritmilor. Un subprogram recursiv presupune că acesta se apelează pe el însuși.

În *Oracle*, o problemă delicată este legată de locul unde se plasează un apel recursiv. De exemplu, dacă apelul este în interiorul unei comenzi *FOR* specifice cursorilor sau între comenzile *OPEN* și *CLOSE*, atunci la fiecare apel este deschis alt cursor. În felul acesta, programul poate depăși limita admisă de cursori deschise la un moment dat (*OPEN_CURSORS*), setată în parametrul de inițializare *Oracle*.

Exemplu:

Să se calculeze recursiv al *m*-lea termen din șirul lui Fibonacci.

```
CREATE OR REPLACE FUNCTION fibo(m POSITIVE) RETURN
INTEGER AS
BEGIN
    IF (m = 1) OR (m = 2) THEN
        RETURN 1;
    ELSE
        RETURN fibo(m-1) + fibo(m-2);
    END IF;
END fibo;
```


Exemplu:

Să se calculeze iterativ al m -lea termen din șirul lui Fibonacci.

```
CREATE OR REPLACE FUNCTION fibo(m POSITIVE) RETURN
INTEGER AS
    ter1 INTEGER := 1;
    ter2 INTEGER := 0;
    valoare INTEGER;
BEGIN
    IF (m = 1) OR (m = 2) THEN
        RETURN 1;
    ELSE
        valoare := ter1 + ter2;
        FOR i IN 3..m LOOP
            ter2 := ter1;
            ter1 := valoare;
            valoare := ter1 + ter2;
        END LOOP;
        RETURN valoare;
    END IF;
END fibo;
```

Declarații *forward*

Subprogramele se numesc reciproc recursive dacă se apelează unul pe altul, în mod direct sau indirect. Declarațiile *forward* permit definirea subprogramelor reciproc recursive.

În *PL/SQL*, un identificator trebuie declarat înainte de a fi folosit. De asemenea, un subprogram trebuie declarat înainte de a fi apelat.

Exemplu:

```
PROCEDURE alfa ( ... ) IS
BEGIN
    beta( ... );          -- apel incorect
    ...
END;
PROCEDURE beta ( ... ) IS
BEGIN
    ...
END;
```

În acest exemplu, procedura *beta* nu poate fi apelată deoarece nu este încă declarată. Problema se poate rezolva simplu în acest caz, inversând ordinea celor două proceduri. Această soluție nu este eficientă întotdeauna (de exemplu, dacă

și procedura *beta* conține un apel al procedurii *alfa*).

PL/SQL permite un tip special, numit *forward*, de declarare a unui subprogram. El constă dintr-o specificare a antetului unui subprogram, terminată prin caracterul „;”. O declarație de tip *forward* pentru procedura *beta* are forma:

```
PROCEDURE beta ( ... );    -- declarație forward
...
PROCEDURE alfa ( ... ) IS
BEGIN
    beta( ... );
...
END;
PROCEDURE beta ( ... ) IS
BEGIN
    ...
END;
```

Declarațiile *forward* pot fi folosite pentru a defini subprograme într-o anumită ordine logică, pentru a defini subprograme reciproc recursive sau pentru a grupa subprograme într-un pachet.

Lista parametrilor formali din declarația *forward* trebuie să fie identică cu cea corespunzătoare corpului subprogramului. Corpul subprogramului poate apărea oriunde după declarația sa *forward*, dar să rămână în aceeași unitate de program.

Utilizarea în expresii *SQL* a funcțiilor definite de utilizator

O funcție stocată poate fi referită într-o comandă *SQL* la fel ca orice funcție standard furnizată de sistem (*built-in function*), dar cu anumite restricții.

Funcțiile *PL/SQL* definite de utilizator pot fi apelate din orice expresie *SQL* în care se pot folosi funcții *SQL* standard.

Funcțiile *PL/SQL* pot să apară în:

- lista de câmpuri a comenzii *SELECT*;
- condiția clauzelor *WHERE* și *HAVING*;
- clauzele *CONNECT BY*, *START WITH*, *ORDER BY* și *GROUP BY*;
- clauza *VALUES* a comenzii *INSERT*;
- clauza *SET* a comenzii *UPDATE*.

Exemplu:

Să se afișeze fotografiile (titlu, valoare, stare) a căror valoare este mai mare decât valoarea medie a tuturor fotografiilor expuse.

```
CREATE OR REPLACE FUNCTION valoare_medie
RETURN NUMBER AS
v_val_mediu fotografie.valoare%TYPE;
```

```

BEGIN
    SELECT AVG(valoare)
    INTO    v_val_mediu
    FROM    fotografie;
    RETURN v_val_mediu;
END;

```

Referirea acestei funcții într-o comandă *SQL* se poate face prin secvența:

```

SELECT titlu, valoare, stare
FROM    fotografie
WHERE   valoare >= valoare_medie;

```

Există restricții referitoare la folosirea funcțiilor definite de utilizator într-o comandă *SQL*. Câteva dintre acestea, care s-au păstrat și pentru *Oracle9i*, vor fi enumerate în continuare:

- funcția definită de utilizator trebuie să fie o funcție stocată (procedurile stocate nu pot fi apelate în expresii *SQL*), nu poate fi locală altui bloc;
- funcția definită de utilizator trebuie să fie o funcție linie și nu una grup (restricția dispare în *Oracle9i*);
- funcția apelată dintr-o comandă *SELECT* sau din comenzi paralelizate *INSERT*, *UPDATE* și *DELETE* nu poate modifica tabelele bazei de date;
- funcția apelată dintr-o comandă *UPDATE* sau *DELETE* nu poate interoga sau modifica tabele ale bazei reactualizate chiar de aceste comenzi (*table mutating*);
- funcția apelată din comenzile *SELECT*, *INSERT*, *UPDATE* sau *DELETE* nu poate conține comenzi *COMMIT*, *ALTER SYSTEM*, *SET ROLE* sau comenzi *CREATE*;
- funcția nu poate apărea în clauza *CHECK* a unei comenzi *CREATE/ALTER TABLE*;
- funcția nu poate fi folosită pentru a specifica o valoare implicită pentru o coloană în cadrul unei comenzi *CREATE/ALTER TABLE*;
- funcția poate fi utilizată într-o comandă *SQL* numai de către proprietarul funcției sau de utilizatorul care are privilegiul *EXECUTE* asupra acesteia;
- parametrii unei funcții *PL/SQL* apelate dintr-o comandă *SQL* trebuie să fie specificați prin poziție (specificarea prin nume nefiind permisă);
- funcția definită de utilizator, apelabilă dintr-o comandă *SQL*, trebuie să aibă doar parametri de tip *IN*, cei de tip *OUT* și *IN OUT* nefiind acceptați;
- parametrii formali ai unui subprogram funcție trebuie să fie de tip

specific bazei de date (*NUMBER*, *CHAR*, *VARCHAR2*, *ROWID*, *LONG*, *LONGROW*, *DATE* etc.) și nu de tipuri *PL/SQL* (*BOOLEAN*, *RECORD* etc.);

- tipul returnat de un subprogram funcție trebuie să fie un tip intern pentru *server*, nu un tip *PL/SQL*;
- funcția nu poate apela un subprogram care nu respectă restricțiile anterioare.

Exemplu:

```
CREATE OR REPLACE FUNCTION calcul (p_val NUMBER)
  RETURN NUMBER IS
BEGIN
  INSERT INTO fotografie(cod_fotografie, gen,
data_achizitie, valoare)
  VALUES (1358, 'arhival', SYSDATE, 700000);
  RETURN (p_val*7);
END;
/

UPDATE fotografie
SET valoare = calcul (550000)
WHERE cod_fotografie = 7531;
```

Comanda *UPDATE* va returna o eroare deoarece tabelul *fotografie* este *mutating*. Operația de reactualizare este însă permisă asupra oricărui alt tabel diferit de *fotografie*.

Informații referitoare la subprograme

Informațiile referitoare la subprogramele *PL/SQL* și modul de acces la aceste informații sunt următoarele:

- codul sursă, utilizând vizualizarea *USER_SOURCE* din dicționarul datelor;
- informații generale, utilizând vizualizarea *USER_OBJECTS* din dicționarul datelor;
- tipul parametrilor (*IN*, *OUT*, *IN OUT*), utilizând comanda *DESCRIBE* din *SQL*Plus*;
- *p-code* (nu este accesibil utilizatorilor);
- erorile la compilare, utilizând vizualizarea *USER_ERRORS* din dicționarul datelor sau comanda *SHOW ERRORS*;
- informații de depanare, utilizând pachetul *DBMS_OUTPUT*.

Atunci când este creat un subprogram stocat, informațiile referitoare la subprogram sunt depuse în dicționarul datelor.

Vizualizarea *USER_OBJECTS* conține informații generale despre toate obiectele prelucrate în baza de date și, în particular, despre subprogramele stocate.

Vizualizarea *USER_OBJECTS* are următoarele câmpuri:

- *OBJECT_NAME* – numele obiectului;
- *OBJECT_TYPE* – tipul obiectului (*PROCEDURE*, *FUNCTION* etc.);
- *OBJECT_ID* – identificator intern al obiectului;
- *CREATED* – data la care a fost creat obiectul;
- *LAST_DDL_TIME* – data ultimei modificări a obiectului;
- *TIMESTAMP* – data și momentul ultimei recompilări;
- *STATUS* – starea de validitate a obiectului.

Pentru a verifica dacă recompilarea explicită (*ALTER*) sau implicită a avut succes se poate verifica starea subprogramelor utilizând coloana *STATUS* din vizualizarea *USER_OBJECTS*.

Orice obiect are o stare (*status*) sesizată în dicționarul datelor, care poate fi:

- *VALID* (obiectul a fost compilat și poate fi folosit atunci când este referit);
- *INVALID* (obiectul trebuie compilat înainte de a fi folosit).

Exemplu:

Să se listeze în ordine alfabetică, procedurile și funcțiile deținute de utilizatorul curent, precum și starea acestora.

```
SELECT      OBJECT_NAME, OBJECT_TYPE, STATUS
FROM        USER_OBJECTS
WHERE       OBJECT_TYPE IN ('PROCEDURE', 'FUNCTION')
ORDER BY    OBJECT_NAME;
```

După ce subprogramul a fost creat, codul sursă al acestuia poate fi obținut consultând vizualizarea *USER_SOURCE* din dicționarul datelor. Vizualizarea are următoarele câmpuri: *NAME* (numele obiectului), *TYPE* (tipul obiectului), *LINE* (numărul liniei din codul sursă), *TEXT* (textul liniilor codului sursă).

Exemplu:

Să se afișeze codul complet pentru funcția *numar_fotografii*.

```
SELECT      TEXT
FROM        USER_SOURCE
WHERE       NAME = 'NUMAR_FOTOGRAFII'
ORDER BY    LINE;
```

Exemplu:

Să se scrie o procedură care recompilează toate obiectele invalide din schema personală.

```
CREATE OR REPLACE PROCEDURE recompileaza IS
  CURSOR obj_curs IS
    SELECT OBJECT_TYPE, OBJECT_NAME
    FROM   USER_OBJECTS
    WHERE  STATUS = 'INVALID'
    AND    OBJECT_TYPE IN
           ('PROCEDURE', 'FUNCTION', 'PACKAGE',
            'PACKAGE BODY', 'VIEW');
BEGIN
  FOR obj_rec IN obj_curs LOOP
    DBMS_DDL.ALTER_COMPILE(obj_rec.OBJECT_TYPE,
                           USER, obj_rec.OBJECT_NAME);
  END LOOP;
END recompileaza;
```

Atunci când se recompilează un obiect *PL/SQL*, *server-ul* va recompila orice obiect invalid de care depinde acesta.

Dacă la recompilarea automată implicită a procedurilor locale dependente apar probleme, atunci starea obiectului va rămâne *INVALID* și *server-ul Oracle* va semnaliza o eroare. Prin urmare:

- este preferabil ca recompilarea să fie manuală, explicită utilizând comanda *ALTER (PROCEDURE, FUNCTION, TRIGGER, PACKAGE)* cu opțiunea *COMPILE*;
- este necesar ca după o schimbare referitoare la obiectele bazei, recompilarea să se facă cât mai repede.

Vizualizarea *USER_ERRORS* afișează textul tuturor erorilor de compilare. Câmpurile acesteia (*NAME*, *TYPE*, *SEQUENCE*, *LINE*, *POSITION*, *TEXT*) sunt analizate în capitolul referitor la tratarea excepțiilor.

Pentru a obține valori (de exemplu, valoarea contorului pentru un *LOOP*, valoarea unei variabile înainte și după o atribuire etc.) și mesaje (de exemplu, părăsirea unui subprogram, apariția unei operații etc.) dintr-un bloc *PL/SQL* pot fi utilizate procedurile pachetului *DBMS_OUTPUT*. Aceste informații se cumulează într-un *buffer* care poate fi consultat ulterior.

Dependența subprogramelor

Atunci când este compilat un subprogram, toate obiectele *Oracle* care sunt referite vor fi înregistrate în dicționarul datelor. Subprogramul este dependent de aceste obiecte. Un subprogram care are erori la compilare este

marcat ca *INVALID* în dicționarul datelor. Un subprogram stocat poate deveni, de asemenea, invalid după execuția unei operații *LDD* asupra unui obiect de care depinde.

<u>Obiecte dependente</u>	<u>Obiecte referite</u>
<i>View</i>	<i>Table</i>
<i>Procedure</i>	<i>View</i>
<i>Function</i>	<i>Procedure</i>
<i>Package Specification</i>	<i>Function</i>
<i>Package Body</i>	<i>Synonym</i>
<i>Database Trigger</i>	<i>Package Specification</i>

Modificarea definiției unui obiect referit poate să influențeze (sau nu) funcționarea normală a obiectului dependent.

Există două tipuri de dependențe:

- dependență directă, în care obiectul dependent (*procedure* sau *function*) face referință direct la un obiect de tip *table*, *view*, *sequence*, *procedure*, *function*;
- dependență indirectă, în care obiectul dependent (*procedure* sau *function*) face referință indirect la un obiect de tip *table*, *view*, *sequence*, *procedure*, *function* prin intermediul unui *view*, *procedure* sau *function*.

În cazul dependențelor locale, atunci când un obiect referit este modificat, obiectele dependente sunt invalidate. La următorul apel al obiectului invalidat, acesta va fi recompilat automat de către *server-ul Oracle*.

În cazul dependențelor la distanță, procedurile stocate local și toate obiectele dependente vor fi invalidate. Ele nu vor fi recompilate automat la următorul apel.

Exemplu:

Se presupune că procedura *filtru* va referi direct tabelul *fotografie* și că procedura *adaug* va reactualiza tabelul *fotografie* prin intermediul unei vizualizări *nou_fotografie*. Pentru aflarea dependențelor directe se poate utiliza vizualizarea *USER_DEPENDENCIES* din dicționarul datelor.

```
SELECT NAME, TYPE, REFERENCED_NAME, REFERENCED_TYPE
FROM   USER_DEPENDENCIES
WHERE  REFERENCED_NAME IN ('fotografie', 'nou_fotografie');
```

<u>NAME</u>	<u>TYPE</u>	<u>REFERENCED_NAME</u>	<u>REFERENCED_TYPE</u>
<i>filtru</i>	<i>Procedure</i>	<i>fotografie</i>	<i>Table</i>
<i>adaug</i>	<i>Procedure</i>	<i>nou_fotografie</i>	<i>View</i>
<i>nou_fotografie</i>	<i>View</i>	<i>fotografie</i>	<i>Table</i>

Dependențele indirecte pot fi afișate utilizând vizualizările *DEPTREE* și

IDEPTREE. Vizualizarea *DEPTREE* afișează o reprezentare a tuturor obiectelor dependente (direct sau indirect). Vizualizarea *IDEPTREE* afișează o reprezentare a aceleași informații, sub forma unui arbore.

Pentru a utiliza aceste vizualizări furnizate de sistemul *Oracle* trebuie:

- executat scriptul *UTLDTREE*;
- executată procedura *DEPTREE_FILL* (are trei argumente: tipul obiectului referit, schema obiectului referit, numele obiectului referit).

Exemplu:

```
@UTLDTREE
EXECUTE DEPTREE_FILL ('TABLE', 'SCOTT', 'fotografie
')
SELECT NESTED_LEVEL, TYPE, NAME
FROM DEPTREE
ORDER BY SEQ#;
```

<u>NESTED_LEVEL</u>	<u>TYPE</u>	<u>NAME</u>
0	Table	fotografie
1	View	nou_fotografie
2	Procedure	adaug
1	Procedure	filtru

```
SELECT *
FROM IDEPTREE;
```

```
DEPENDENCIES
TABLE nume_schema.fotografie
VIEW nume_schema.nou_fotografie
PROCEDURE nume_schema.adaug
PROCEDURE nume_schema.filtru
```

Dependențele la distanță sunt tratate printr-una dintre modalitățile alese de utilizator, modelul *timestamp* (implicit) sau modelul *signature*.

Fiecare unitate *PL/SQL* are un *timestamp* (etichetă de timp) care este setat atunci când unitatea este creată sau recompilată și care este depus în câmpul *LAST_DDL_TIME* din dicționarul datelor. Modelul *timestamp* realizează compararea momentelor ultimei modificări a celor două obiecte analizate. Dacă obiectul bazei are momentul ultimei modificări mai recent decât cel al obiectului dependent, atunci obiectul dependent va fi recompilat.

Modelul *signature* (semnătură) determină momentul la care obiectele bazei distante trebuie recompilate. Când este creat un subprogram, o *signature* este depusă în dicționarul datelor, alături de *p-code*. Aceasta conține: numele

construcției *PL/SQL* (*PROCEDURE*, *FUNCTION*, *PACKAGE*), tipurile parametrilor, ordinea parametrilor, numărul acestora și modul de transmitere (*IN*, *OUT*, *IN OUT*). Dacă parametrii se schimbă, atunci evident *signature* se schimbă.

Pentru a folosi modelul *signature* este necesară setarea parametrului *REMOTE_DEPENDENCIES_MODE* la *SIGNATURE*. Aceasta se poate realiza prin:

- 1) comanda *ALTER SESSION*, care va afecta doar sesiunea curentă:

ALTER SESSION SET REMOTE_DEPENDENCIES_MODE = SIGNATURE;

- 2) comanda *ALTER SYSTEM*, care va afecta întreaga bază de date (toate sesiunile), dar trebuie avut privilegiul sistem pentru a utiliza această instrucțiune:

ALTER SYSTEM SET REMOTE_DEPENDENCIES_MODE = SIGNATURE;

- 3) adăugarea în fișierul de inițializare a unei linii de forma:

REMOTE_DEPENDENCIES_MODE = SIGNATURE;

Recompilarea procedurilor și a funcțiilor dependente este fără succes dacă:

- obiectul referit este suprimat (*DROP*) sau redenumit (*RENAME*);
- tipul coloanei referite este schimbat;
- o vizualizare referită este înlocuită printr-o vizualizare ce conține alte coloane;
- lista parametrilor unei proceduri referite este modificată.

Recompilarea procedurilor și funcțiilor dependente este cu succes dacă:

- tabelul referit are noi coloane;
- corpul *PL/SQL* al unei proceduri referite a fost modificat și recompilat cu succes.

Erorile datorate dependențelor pot fi minimizate:

- utilizând comenzi *SELECT* cu opțiunea „*“;
- incluzând lista coloanelor în comanda *INSERT*;
- declarând variabile cu atributul *%TYPE*;
- declarând înregistrări cu atributul *%ROWTYPE*;

Concluzii:

- Dacă procedura depinde de un obiect local, atunci se face recompilare automată la prima reexecuție.
- Dacă procedura depinde de o procedură distantă, atunci se face recompilare automată, dar la a doua reexecuție. Este preferabilă o recompilare manuală pentru prima reexecuție sau implementarea unei

strategii de reinvocare a ei (a doua oară).

- Dacă procedura depinde de un obiect distant, dar care nu este procedură, atunci nu se face recompilare automată.

Rutine externe

PL/SQL a fost special conceput pentru *Oracle* și este specializat pentru procesarea tranzacțiilor *SQL*. Totuși, într-o aplicație complexă pot să apară cerințe și funcționalități care sunt mai eficient de implementat în *C*, *Java* sau alt limbaj de programare. De exemplu, *Java* este un limbaj portabil cu un model de securitate bine definit, care lucrează excelent pentru aplicații *Internet*.

Dacă aplicația trebuie să efectueze anumite acțiuni care nu pot fi implementate optim utilizând *PL/SQL*, atunci este preferabil să fie utilizate alte limbaje care realizează performant acțiunile respective. În acest caz este necesară comunicarea între diferite module ale aplicației care sunt scrise în limbaje diferite.

Până la versiunea *Oracle8*, modalitatea de comunicare între *PL/SQL* și alte limbaje (de exemplu, limbajul *C*) a fost utilizarea pachetelor *DBMS_PIPE* și/sau *DBMS_ALERT*.

Începând cu *Oracle8*, comunicarea este simplificată prin utilizarea rutinelor externe. O rutină externă este o procedură sau o funcție scrisă într-un limbaj diferit de *PL/SQL*, dar apelabilă dintr-un program scris în *PL/SQL*. *PL/SQL* extinde funcționalitatea server-ului *Oracle*, furnizând o interfață pentru apelarea rutinelor externe. Orice bloc *PL/SQL* executat pe *server* sau pe *client* poate apela o rutină externă. Singurul limbaj acceptat pentru rutine externe în *Oracle8* este limbajul *C*.

Pentru a marca apelarea unei rutine externe în programul *PL/SQL* este definit un punct de intrare (*wrapper*) care direcționează spre codul extern (program *PL/SQL* → *wrapper* → cod extern). Pentru crearea unui *wrapper* este utilizată o clauză specială (*AS EXTERNAL*) în cadrul comenzii *CREATE OR REPLACE PROCEDURE*. De fapt, clauza conține informații referitoare la numele bibliotecii în care se găsește subprogramul extern (clauza *LIBRARY*), numele rutinei externe (clauza *NAME*) și corespondența (*C* ↔ *PL/SQL*) dintre tipurile de date (clauza *PARAMETERS*). În ultimele versiuni s-a renunțat la clauza *AS EXTERNAL*.

Rutinele externe (scrise în *C*) sunt compilate, apoi depuse într-o bibliotecă dinamică (*DLL* – *dynamic link library*) și sunt încărcate doar atunci când este necesar. Dacă se invocă o rutină externă scrisă în *C*, trebuie setată conexiunea spre această rutină. Un proces numit *extproc* este declanșat automat de către *server*. La rândul său, procesul *extproc* va încărca biblioteca identificată prin clauza *LIBRARY* și va apela rutina respectivă.

Oracle8i permite utilizarea de rutine externe scrise în *Java*. De asemenea, un *wrapper* poate include specificații de apelare, prin utilizarea clauzei *LANGUAGE*. De fapt, aceste specificații permit apelarea rutinelor externe scrise în orice limbaj. De exemplu, o procedură scrisă într-un limbaj diferit de *C* sau *Java* poate fi utilizată în *SQL* sau *PL/SQL* dacă procedura respectivă este apelabilă din *C*. În felul acesta, biblioteci standard scrise în alte limbaje de programare pot fi apelate din programe *PL/SQL*.

Procedura *PL/SQL* executată pe *server*-ul *Oracle* poate apela o rutină externă scrisă în limbajul *C*, depusă într-o bibliotecă partajată. Procedura *C* se execută într-un spațiu adresă diferit de cel al *server*-ului *Oracle*, în timp ce unitățile *PL/SQL* și metodele *Java* se execută în spațiul adresă al *server*-ului. *JVM* (*Java Virtual Machine*) de pe *server* va executa metoda *Java* în mod direct, fără a fi necesar procesul *extproc*.

Maniera de încărcare depinde de limbajul în care este scrisă rutina.

- Pentru a apela rutine externe *C*, *server*-ul trebuie să cunoască poziționarea bibliotecii dinamice *DLL*. Acest lucru este furnizat de *alias*-ul bibliotecii din clauza *AS LANGUAGE*.
- Pentru apelarea unei rutine externe *Java* se va încărca clasa *Java* în baza de date. Este necesară doar crearea unui *wrapper* care direcționează către codul extern. Spre deosebire de rutinele externe *C*, nu este necesară nici bibliotecă și nici setarea conexiunii spre rutina externă.

Clauza *LANGUAGE* din cadrul comenzii de creare a unui subprogram, specifică limbajul în care este scrisă rutina (procedură externă *C* sau metodă *Java*) și are următoarea formă:

{IS | AS} LANGUAGE {C | JAVA}

Pentru o procedură *C* sunt date informații referitoare la numele acesteia (clauza *NAME*); *alias*-ul bibliotecii în care se găsește (clauza *LIBRARY*); opțiuni referitoare la tipul, poziția, lungimea, modul de transmitere (prin valoare sau prin referință) al parametrilor (clauza *PARAMETERS*); posibilitatea ca rutina externă să acceseze informații despre parametri, excepții, alocarea memoriei utilizator (clauza *WITH CONTEXT*).

LIBRARY nume_biblioteca [**NAME** nume_proc_c] [**WITH CONTEXT**]
[PARAMETERS (parametru_extern [, parametru_extern ...])]

Pentru o metodă *Java*, în clauză trebuie specificată doar semnatura metodei (lista tipurilor parametrilor în ordinea apariției).

Exemplu:

```
CREATE OR REPLACE FUNCTION calc (x IN REAL) RETURN
NUMBER
AS LANGUAGE C
    LIBRARY biblioteca
```

```
NAME "c_calc"
PARAMETERS (x BY REFERENCES);
```

O rutină externă nu este apelată direct, ci se apelează subprogramul *PL/SQL* care referă rutina externă. Apelarea poate să apară în: blocuri anonime, subprograme independente sau care aparțin unui pachet, metode ale unui tip obiect, declanșatori bază de date, comenzi *SQL* care apelează funcții (în acest caz, trebuie utilizată clauza *PRAGMA RESTRICT_REFERENCES*).

De remarcat că o metodă *Java* poate fi apelată din orice bloc *PL/SQL*, subprogram sau pachet. *JDBC* (*Java Database Connectivity*), care reprezintă interfața *Java* standard pentru conectare la baze de date relaționale, și *SQLJ* permit apelarea de blocuri *PL/SQL* din programe *Java*. *SQLJ* face posibilă incorporarea operațiilor *SQL* în codul *Java*. Standardul *SQLJ* acoperă doar operații *SQL* statice. *Oracle9i SQLJ* include extensii pentru a suporta direct *SQL* dinamic.

O altă modalitate de a încărca metode *Java* este folosirea interactivă în *SQL*Plus* a comenzii *CREATE JAVA instrucțiune*.

Funcții tabel

O funcție tabel (*table function*) returnează drept rezultat un set de linii (de obicei, sub forma unei colecții). Această funcție poate fi interogată direct printr-o comandă *SQL*, ca și cum ar fi un tabel al bazei de date. În felul acesta, funcția poate fi utilizată în clauza *FROM* a unei cereri.

O funcție tabel conductă (*pipelined table function*) este similară unei funcții tabel, dar returnează datele iterativ, pe măsură ce acestea sunt obținute, nu toate deodată. Aceste funcții sunt mai eficiente deoarece informația este returnată imediat cum este obținută.

Conceptul de funcție tabel conductă a fost introdus în versiunea *Oracle9i*. Utilizatorul poate să definească astfel de funcții. De asemenea, este posibilă execuția paralelă a funcțiilor tabel (evident și a celor clasice). În acest caz, funcția trebuie să conțină în declarație opțiunea *PARALLEL_ENABLE*.

Funcția tabel conductă acceptă orice argument pe care îl poate accepta o funcție obișnuită și trebuie să returneze o colecție (*nested table* sau *varray*). Ea este declarată specificând cuvântul cheie *PIPELINED* în comanda *CREATE OR REPLACE FUNCTION*. Funcția tabel conductă trebuie să se termine printr-o comandă *RETURN* simplă, care nu întoarce nici o valoare.

Pentru a returna un element individual al colecției este folosită comanda *PIPE ROW*, care poate să apară numai în corpul unei funcții tabel conductă, în caz contrar generându-se o eroare. Comanda poate fi omisă dacă funcția tabel conductă nu returnează nici o linie.

După ce funcția a fost creată, ea poate fi apelată dintr-o cerere *SQL*

utilizând operatorul *TABLE*. Cererile referitoare la astfel de funcții pot să includă cursoare și referințe la cursoare, respectându-se semantica de la cursoarele clasice.

Funcția tabel conductă nu poate să apară în comenzile *INSERT*, *UPDATE*, *DELETE*. Totuși, pentru a realiza o reactualizare, poate fi creată o vizualizare relativă la funcția tabel și folosit un declanșator *INSTEAD OF*.

Exemplu:

Să se obțină o instanță a unui tabel ce conține informații referitoare la denumirea zilelor săptămânii.

Problema este rezolvată în două variante. Prima reprezintă o soluție clasică, iar a doua variantă implementează problema cu ajutorul unei funcții tabel conductă.

Varianta 1:

```
CREATE TYPE t_linie AS OBJECT (
    id1 NUMBER, sir VARCHAR2(20));
CREATE TYPE t_tabel AS TABLE OF t_linie;
CREATE OR REPLACE FUNCTION calc1 RETURN t_tabel
AS
    v_tabel t_tabel;
BEGIN
    v_tabel := t_tabel (t_linie (1, 'luni'));
    FOR j IN 2..7 LOOP
        v_tabel.EXTEND;
        IF j = 2
            THEN v_tabel(j) := t_linie (2, 'marti');
        ELSIF j = 3
            THEN v_tabel(j) := t_linie (3, 'miercuri');
        ELSIF j = 4
            THEN v_tabel(j) := t_linie (4, 'joi');
        ELSIF j = 5
            THEN v_tabel(j) := t_linie (5, 'vineri');
        ELSIF j = 6
            THEN v_tabel(j) := t_linie (6, 'sambata');
        ELSIF j = 7
            THEN v_tabel(j) := t_linie (7, 'duminica');
        END IF;
    END LOOP;
    RETURN v_tabel;
END calc1;
```

Funcția *calc1* poate fi invocată în clauza *FROM* a unei comenzi *SELECT*:

30

```
SELECT *  
FROM TABLE (CAST (calc1 AS t_tabel));
```

Varianta 2:

```
CREATE OR REPLACE FUNCTION calc2 RETURN t_tabel  
PIPELINED  
AS  
    v_linie t_linie;  
BEGIN  
    FOR j IN 1..7 LOOP  
        v_linie :=  
            CASE j  
                WHEN 1 THEN t_linie (1, 'luni')  
                WHEN 2 THEN t_linie (2, 'marti')  
                WHEN 3 THEN t_linie (3, 'miercuri')  
                WHEN 4 THEN t_linie (4, 'joi')  
                WHEN 5 THEN t_linie (5, 'vineri')  
                WHEN 6 THEN t_linie (6, 'sambata')  
                WHEN 7 THEN t_linie (7, 'duminica')  
            END;  
        PIPE ROW (v_linie);  
    END LOOP;  
    RETURN;  
END calc2;
```

Se observă că tabelul este implicat doar în tipul rezultatului. Pentru apelarea funcției *calc2* este folosită sintaxa următoare:

```
SELECT *  
FROM TABLE (calc2);
```

Funcțiile tabel sunt folosite frecvent pentru conversii de tipuri de date. *Oracle9i* introduce posibilitatea de a crea o funcție tabel care returnează un tip *PL/SQL* (definit într-un bloc). Funcția tabel care furnizează (la nivel de pachet) drept rezultat un tip de date trebuie să fie de tip conductă. Pentru apelare este utilizată sintaxa simplificată (fără *CAST*).

Exemplu:

```
CREATE OR REPLACE PACKAGE exemplu IS  
    TYPE t_linie IS RECORD  
        (idl NUMBER, sir VARCHAR2(20));  
    TYPE t_tabel IS TABLE OF t_linie;  
END exemplu;  
CREATE OR REPLACE FUNCTION calc3  
RETURN exemplu.t_tabel
```

31

```
PIPELINED AS
v_linie  exemplu.t_linie;
BEGIN
  FOR j IN 1..7 LOOP
    CASE j
      WHEN 1 THEN v_linie.idl := 1;
v_linie.sir := 'luni';
      WHEN 2 THEN v_linie.idl := 2;
v_linie.sir := 'marti';
      WHEN 3 THEN v_linie.idl := 3;
v_linie.sir := 'miercuri';
      WHEN 4 THEN v_linie.idl := 4;
v_linie.sir := 'joi';
      WHEN 5 THEN v_linie.idl := 5;
v_linie.sir := 'vineri';
      WHEN 6 THEN v_linie.idl := 6;
v_linie.sir := 'sambata';
      WHEN 7 THEN v_linie.idl := 7;
v_linie.sir := 'duminica';
    END CASE;
    PIPE ROW (v_linie);
  END LOOP;
  RETURN;
END calc3;
```

Procesarea tranzacțiilor autonome

Tranzacția este o unitate logică de lucru, adică o secvență de comenzi care trebuie să se execute ca un întreg pentru a menține consistența bazei de date. În mod uzual, o tranzacție poate să cuprindă mai multe blocuri, iar într-un bloc pot să fie mai multe tranzacții.

O tranzacție autonomă este o tranzacție independentă lansată de altă tranzacție, numită tranzacție principală. Tranzacția autonomă permite suspendarea tranzacției principale, executarea de comenzi *SQL*, permanentizarea (*commit*) și anularea (*rollback*) acestor operații.

Odată începută, tranzacția autonomă este independentă, în sensul că nu partajează blocări, resurse sau dependențe cu tranzacția principală. În felul acesta, o aplicație nu trebuie să cunoască operațiile autonome ale unei proceduri, iar procedura nu trebuie să cunoască nimic despre tranzacțiile aplicației. Tranzacția autonomă are totuși toate funcționalitățile unei tranzacții obișnuite (permite cereri paralele, procesări distribuite etc.).

Pentru definirea unei tranzacții autonome se utilizează clauza *PRAGMA AUTONOMOUS_TRANSACTION* care informează compilatorul *PL/SQL* că trebuie să marcheze o rutină ca fiind autonomă. Prin rutină se înțelege: bloc anonim de cel mai înalt nivel (nu imbricat); procedură sau funcție locală, independentă sau împachetată; metodă a unui tip obiect; declanșator bază de date.

Codul *PRAGMA AUTONOMOUS_TRANSACTION* se specifică în partea declarativă a rutinei.

Codul *PRAGMA AUTONOMOUS_TRANSACTION* marchează numai rutine individuale ca fiind independente. Nu pot fi marcate toate subprogramele unui pachet sau toate metodele unui tip obiect ca autonome. Prin urmare, clauza nu poate să apară în partea de specificație a unui pachet.

Observații:

- Declanșatorii autonomi, spre deosebire de cei clasici pot conține comenzi *LCD* (de exemplu, *COMMIT*, *ROLLBACK*).
- Excepțiile declanșate în cadrul tranzacțiilor autonome generează un *rollback* la nivel de tranzacție și nu la nivel de instrucțiune.
- Când se intră în secțiunea executabilă a unei tranzacții autonome, tranzacția principală este suspendată.

Cu toate că o tranzacție autonomă este inițiată de altă tranzacție, ea nu este o tranzacție imbricată deoarece:

- nu partajează resurse cu tranzacția principală;
- nu depinde de tranzacția principală (de exemplu, dacă tranzacția principală este anulată, atunci tranzacțiile imbricate sunt de asemenea anulate, în timp ce tranzacția autonomă nu este anulată);
- schimbările permanentizate din tranzacții autonome sunt vizibile imediat altor tranzacții, pe când cele din tranzacții imbricate sunt vizibile doar după ce tranzacția principală este permanentizată.